Box Plot

```
pBox(M#):=
  [ M:=eval(M#) P:=0 A:=eval(matrix(0,1)) B:=A ]
  Fd(x):=
    n:=[1..length(x)]
    if (ans:=findrows(eval(augment(n,x)),1,2))=0
      0
    else
      col(ans,1)                                              Find utility
  Md(x):=
    n:=length(x)
    if trunc(0.5·n)=0.5·n
      0.5·(x_(0.5·n) + x_(0.5·n+1))                           Median
    else
      x_(0.5·(n+1))
  for k∈[1..cols(M)]
    if length(y:=col(M,k))=1
      y:=sort(y_1)
    else
      y:=sort(y)
    [A B]:=eval([min(stack(A,min(y))) max(stack(B,max(y)))])
    [Q2:=Md(y) s:=6·(k−1)]                                    Quartil
    [ if (F:=Fd(y<Q2))=0 if (F:=Fd(y>Q2))=0
        Q1:=Q2                Q3:=Q2
      else                  else
        Q1:=Md(y_F)           Q3:=Md(y_F) ]
    O:=Fd((y<Q1−1.5·(Q3−Q1))∨(y>Q3+1.5·(Q3−Q1)))              Outliers
    if O=0
      P_(s+3):=""
    else
      P_(s+3):=augment(k−0.5, y_O, "x")
      y_O:=Q2
    [a b]:=eval([min(y) max(y)])
    P_(s+1):=eval(stack([[k−0.7 a],[k−0.3 a],[k−0.5 a],[k−0.5 Q1],[k−0.8 Q1],[k−0.8 Q3],[k−0.5 Q3]],[k−0.5 b],[[k−0.7 b],[k−0.3 b],[k−0.5 b],[k−0.5 Q3],[k−0.2 Q3],[k−0.2 Q1],[k−0.5 Q1]]))   Boxes
    P_(s+2):=[k−0.8 Q2; k−0.2 Q2]
    P_(s+[4..6]):=""
  p:=0.05·(B−A)
  str2num(concat("sys",substr(num2str(P_(s+6):=[0 A−p; 0 B+p; k B+p; k A−p; 0 A−p]),4)))
```

```
pBox5(M) := | P := str2num(concat("mat", substr(num2str(pBox(M)), 4)))
            | F := ["min" "q1" "q2" "q3" "max"]
            | for k ∈ [1 .. rows(P)/6]
            |     f(m, n) := P[6·(k−1)+m, n]_2
            |     F := eval(stack(F, [f(1,1) f(1,4) f(2,1) f(1,6) f(1,8)]))
            | F
```

Examples

```
rnd(a, b) := a + (b − a)·10^−9 · random(10^9)
X := matrix(300, 10)
[n := [1 .. length(X)] X := eval(X_n := rnd(−2, 12))]
Add some outliers
[n := [1 .. 20] X := eval(X_n := rnd(−16, −2))]
[n := [21 .. 40] X := eval(X_n := rnd(12, 26))]
```



pBox(*X*)

$$pBox5(X) = \begin{bmatrix} \text{"min"} & \text{"q1"} & \text{"q2"} & \text{"q3"} & \text{"max"} \\ -8.56 & 1.46 & 4.97 & 8.37 & 11.97 \\ -2.5 & 2.41 & 6.14 & 8.94 & 11.97 \\ -6.9 & 1.31 & 5.14 & 8.38 & 14.04 \\ -2.8 & 1.11 & 4.83 & 8.52 & 15.47 \\ -6.16 & 1.11 & 4.41 & 8.04 & 11.94 \\ -2.41 & 1.45 & 5.34 & 8.18 & 13.55 \\ -8.43 & 1.22 & 5 & 7.99 & 11.8 \\ -7.44 & 1.9 & 5.63 & 9.17 & 18.39 \\ -4.87 & 1.03 & 4.8 & 8.47 & 18.39 \\ -1.99 & 1.96 & 5.23 & 8.85 & 14.42 \end{bmatrix}$$

pBox5 returns the 'five numbers summary' of the data, where min & max are without outliers.

It also accept data with different lengths. To do this, save the data to be plotted in each line of a vector.

```
n := [1 .. cols(X)]
M_{1 n} := X_[1 .. rnd(100, 150)]_n
```

$$\overrightarrow{rows(M)} = [106\ 144\ 138\ 116\ 149\ 135\ 147\ 123\ 148\ 137]$$



pBox(*M*)

$$pBox5(M) = \begin{bmatrix} \text{"min"} & \text{"q1"} & \text{"q2"} & \text{"q3"} & \text{"max"} \\ -2.29 & 1.78 & 5.13 & 8.18 & 11.84 \\ -2.5 & 2.4 & 6.21 & 9.41 & 11.97 \\ -6.9 & 0.68 & 4.93 & 8.62 & 14.04 \\ -2.8 & 2.01 & 5.54 & 9.04 & 15.47 \\ -6.16 & 0.36 & 3.82 & 7.73 & 11.91 \\ -2.41 & 1.49 & 5.4 & 8.37 & 13.55 \\ -8.43 & 1.4 & 5.26 & 8.26 & 11.8 \\ -3.37 & 2.58 & 5.78 & 9.01 & 18.39 \\ -4.87 & 0.8 & 4.01 & 7.88 & 18.39 \\ -1.94 & 1.8 & 4.48 & 8.47 & 14.42 \end{bmatrix}$$

Boxplot of Michelson's data: speed of light - 299000 km/s vs Experiment. Each experiment have 20 runs.

$$MM := \begin{bmatrix} 850 & 960 & 880 & 890 & 890 \\ 740 & 940 & 880 & 810 & 840 \\ 900 & 960 & 880 & 810 & 780 \\ 1070 & 940 & 860 & 820 & 810 \\ 930 & 880 & 720 & 800 & 760 \\ 850 & 800 & 720 & 770 & 810 \\ 950 & 850 & 620 & 760 & 790 \\ 980 & 880 & 860 & 740 & 810 \\ 980 & 900 & 970 & 750 & 820 \\ 880 & 840 & 950 & 760 & 850 \\ 1000 & 830 & 880 & 910 & 870 \\ 980 & 790 & 910 & 920 & 870 \\ 930 & 810 & 850 & 890 & 810 \\ 650 & 880 & 870 & 860 & 740 \\ 760 & 880 & 840 & 880 & 810 \\ 810 & 830 & 840 & 720 & 940 \\ 1000 & 800 & 850 & 840 & 950 \\ 1000 & 790 & 840 & 850 & 800 \\ 960 & 760 & 840 & 850 & 810 \\ 960 & 800 & 840 & 780 & 870 \end{bmatrix}$$

$$co := \frac{c}{\frac{km}{s}} - 299000$$

Alvaro



$$\begin{cases} pBox\ (MM) \\ "" \\ "" \\ co \end{cases}$$

$$pBox5\ (MM) = \begin{bmatrix} "min" & "q1" & "q2" & "q3" & "max" \\ 740 & 850 & 940 & 980 & 1070 \\ 760 & 800 & 845 & 890 & 960 \\ 840 & 840 & 855 & 880 & 910 \\ 720 & 765 & 815 & 870 & 920 \\ 740 & 780 & 810 & 870 & 950 \end{bmatrix}$$